

Jによる時系列解析 (Time Series)

M.Shimura

JCD02773@nifty.ne.jp

2003年8月13日

目次

1	スムージング	3
1.1	単純移動平均	3
1.2	加重移動平均	5
1.3	12ヶ月移動平均	5
1.4	Spencer 移動平均	6
2	差分	11
2.1	経過と解説	11
3	自己相関	14
3.1	自己相関 ACF(k) autocorreration function	14
3.2	C.A. Reiter による自己相関のスク립ト	17
3.3	偏自己相関係数 Pearsons autocorrelation coefficient PACF	18
4	Autoregression 自己回帰	21
4.1	Yule-Walker 法	21
4.2	AIC	23
4.3	回帰と予測	24
5	レビンソンのアルゴリズム	26
5.1	一変数	26

5.2	Householder 法	29
6	自己回帰移動平均 : ARMA	31

1 スムージング

1.1 単純移動平均

移動平均 Moving Average	任意の移動平均	m mav n m 次数 n Data
------------------------	---------	---------------------------

J は、移動平均を次の簡潔なイディオムで表現できる。

```
mav=: +/\ % [
```

1.1.1 経過と解説

J の infix (\) の機能は強力であり左引数 (x.) で指定した数での移動平均のためのずらした組み合わせを簡単に作る。

```
] s=. 10 ? 30
```

```
26 7 15 12 20 23 28 24 1 29
```

```
3 <\ s NB. partition off each 3 (overlapped 2)
```

```
+-----+-----+-----+-----+-----+-----+-----+-----+
|26 7 15|7 15 12|15 12 20|12 20 23|20 23 28|23 28 24|28 24 1|24 1 29|
+-----+-----+-----+-----+-----+-----+-----+-----+
```

```
3 +/\ s
```

```
48 34 47 55 71 75 53 54
```

```
3 mav s
```

```
16 11.3333 15.6667 18.3333 23.6667 25 17.6667 18
```

1.1.2 偶数の x.

移動平均は, x. (左パラメーター) が偶数の場合には, 原データとの対比のために中心化がよく行われる。@. 以下の条件式で偶数奇数の判定を行い、偶数の場合は, 中心化を行っている。

```
mav=:+/\ % [  
mav_c=: 2&mav@mav ' mav @. (2&|@[]) NB.centred MA
```

```
a=.10?.20
```

```
a,:(2 mav a),:2 mav_c a
```

```
17    4    9    7    12    14    18    13    0    6  
10.5  6.5  8    9.5  13    16    15.5  6.5  3    0  
8.5   7.25 8.75 11.25 14.5  15.75 11    4.75 0    0
```

1.1.3 マトリクスデータへの適用

この mav はランクの指定で, マトリクスデータにも一括して適用できる。

*1

```
s2=: ?. 5 12 $ 30
```

```
s2
```

```
3 22 13 15 6 1 20 20 28 11 15 24  
1 1 15 20 0 11 2 12 20 17 27 25  
15 2 19 12 21 27 22 7 1 22 9 18  
22 29 10 7 29 21 22 19 2 18 26 8  
13 22 14 7 8 10 4 14 26 27 1 27
```

```
6.2 ": 3 mav "1 s2
```

*1 6.2": は合計 6 桁、小数点以下 2 桁の切り捨てで文字化して表示

12.67 16.67 11.33 7.33 9.00 13.67 22.67 19.67 18.00 16.67
 5.67 12.00 11.67 10.33 4.33 8.33 11.33 16.33 21.33 23.00
 12.00 11.00 17.33 20.00 23.33 18.67 10.00 10.00 10.67 16.33
 20.33 15.33 15.33 19.00 24.00 20.67 14.33 13.00 15.33 17.33
 16.33 14.33 9.67 8.33 7.33 9.33 14.67 22.33 18.00 18.33

縦方向への移動平均に *vmav* も用意した。

6.2 " : 3 vmav s2

6.33 8.33 15.67 15.67 9.00 13.00 14.67 13.00 16.33 16.67 17.00 22.33
 12.67 10.67 14.67 13.00 16.67 19.67 15.33 12.67 7.67 19.00 20.67 17.00
 16.67 17.67 14.33 8.67 19.33 19.33 16.00 13.33 9.67 22.33 12.00 17.67

1.2 加重移動平均

1.2.1 一次

3ヶ月の移動平均値を更に移動平均する場合。一層のスムージングがはかれる。

$$\hat{Q} = \frac{\frac{Q_{t-2} + Q_{t-1} + Q_t}{3} \frac{Q_{t-1} + Q_t + Q_{t+1}}{3} \frac{Q_t + Q_{t+1} + Q_{t+2}}{3}}{3}$$

$$= \frac{1}{9}Q_{t-2} + \frac{2}{9}Q_{t-1} + \frac{3}{9}Q_t + \frac{2}{9}Q_{t+1} + \frac{1}{9}Q_{t+2}$$

$$= \frac{1}{9}[1 \quad 2 \quad 3 \quad 2 \quad 1]$$

これは、上記のウェイトによる5ヶ月の加重移動平均と同じである。四季の変動の特徴をよく捕らえることができる。

1.3 12ヶ月移動平均

mav12 は単項動詞。 *mav* の左引数 1 2 をに固定した上で中心化を行っている。

mav =: +/\ % [

NB. Season adjustment using 12 month

mav12 =: [: 2&*mav* 12&*mav*

```

2<\ 12 mav i. 24
+-----+-----+-----+-----+-----+-----+-----+-----+
|5.5 6.5|6.5 7.5|7.5 8.5|8.5 9.5|9.5 10.5|10.5 11.5|11.5 12.5|12.5 13.5|
+-----+-----+-----+-----+-----+-----+-----+-----+...
      mav12 i. 24
6 7 8 9 10 11 12 13 14 15 16 17

```

1.4 Spencer 移動平均

スペンサーの15ポイント移動平均
ウエイトを次のように置いており、隣接に重心をおくので、三角形や低位の数値が破壊され
れないで残っている。

これは、次のウエイトに依っている。

$$\frac{1}{4} \begin{bmatrix} 1 & -3 & 3 & 4 & 3 & -3 \end{bmatrix}$$

$$\frac{1}{5} \begin{bmatrix} 1 & 1 & 1 & 1 & 1 \end{bmatrix}$$

$$\frac{1}{4} \begin{bmatrix} 1 & -3 & 3 & 4 & 3 & -3 \end{bmatrix}$$

$$\frac{1}{4} \begin{bmatrix} 1 & -3 & 3 & 4 & 3 & -3 \end{bmatrix}$$

```
wts=: (|. ,}. ) 74 67 46 21 3 _5 _6 _3%320
```

```
spen walk
```

次の例は、東北自動車道那須付近の本線交通量（下り）の6年間の月平均の原データと
スペンサー移動平均を示した。

1.4.1 Script

```
NB. =====
```

```
NB. C.Reiter Fractals Visuarization and J (2nd Edition 2000)
```

```
NB. =====
```

```
cos=: 2&o.
```




図 1: Spencer スペンサー移動平均




図 2: 東北道那須付近の月平均交通量

```

randunif=: (?%<:.)@:( $\$2147483647$ ) : ( {.@[+({:-{.})@[*$:@])
randsn=: cos@+:@o.@randunif * %:@-@+:@^.@randunif

wts=: (|. ,}. ) 74 67 46 21 3 _5 _6 _3%320
locspen=: (+/ . *)&wts
spencer=: 15"_ locspen\ (7: # {.),] ,7:#{:
walk=: +/\ randsn 100
spen=: 3 : 0
NB. Usage spen ''
NB. Usage plot walk// plot spencer walk

```

```
pd 'reset'  
pd 'type line'  
pd 'itemcolor 255 0 0'  
pd walk  
pd 'type line'  
pd 'itemcolor 0 0 255'  
pd spencer walk  
pd 'show'  
)
```


1.4.2 参考文献

Gareth Janacek [Practical Time series] Arnold 2001

C.Reiter [Fractal Visualization ans J 2nd ed.]

1.4.3 移動平均の PLOT

移動平均 Moving Analsys	plotcomb [任意の移動平均のグラフ]	m plotcomb n m 次数 n Data
------------------------	------------------------	--------------------------------

移動平均を求め原データと同時にグラフにあらわす関数を用意した.

- ・ 任意の次数（例では3 = 四半期）と原系列を **plot** する
- 偶数の場合はグラフにあらわすため中心化の処理を行っている。




図 3: 3 月移動平均

1.4.4 Script

```
NB. =====
NB. mav any dimension
NB. usage: x. plotcomb y. (x. dimension y. data)
NB. =====
p_sub0=: 4 : 0
NB. make mav anytime odd or even
mav=: +/\ % [
mav_even=:[: 2&mav mav
select. tmp0 =:2 | x.
case. 0 = tmp0 do. goto_odd.
case. 1 = tmp0 do. goto_even.
end.
label_odd.
tmp10=:x. mav y.
tmp2=:<. -: x.
return.
label_even.
tmp10=: x. mav_even y.
tmp2=:>. -: x.
return.
)
p_sub1=: 4 : 0
NB. make plot data by mav
x. p_sub0 y.
tmp30=(tmp2 # tmp2) , tmp2 +i. # tmp10
tmp3=:tmp30,tmp2 #({: tmp30) NB. x axis
tmp4=( tmp2 # {. tmp10), tmp10,tmp20=:tmp2 # {: tmp10 NB. y axis
```

```

tmp5=:tmp3;tmp4
)

plotcomb=: 4 : 0
NB. usage: x. plotcomb y.
NB. x. jisuu
NB. y. data
pd 'reset'
pd x. p_sub1 y.
pd y.
pd 'color red'
pd tmp5
pd 'show'
)

```

2 差分

差分	<pre> diff=: 3 : '-/(L:0) 2 <\ y.' NB. take difference </pre>	
----	---	--

非定常時系列は一階差分を取ると定常時系列となることが多い。一階差分も `infix` を用いると用意に作れる。

2.1 経過と解説

シードを固定した乱数

```

10 ? 10
8 2 4 3 7 5 1 0 9 6

```

<\ は Infix オーバーラップさせて 2 個ずつ組み合わせる。

```
2<\ 10 ?. 10
+---+---+---+---+---+---+---+---+---+
|8 2|2 4|4 3|3 7|7 5|5 1|1 0|0 9|9 6|
+---+---+---+---+---+---+---+---+---+
```

ボックスの中での除算

```
-/ (L:0) 2<\ 10 ?. 10
+---+---+---+---+---+---+
|6|_2|1|_4|2|4|1|_9|3|
+---+---+---+---+---+---+
```

```
plot > df 100 ?. 100
```




図 4: 差分

```
NB. =====
NB. diff and plot
NB. Usage: diff_plot y.
NB. =====
```

```
diff=:3 : '-/(L:0) 2<\ y.'  
diff_plot=: 3 : 0  
pd 'reset'  
pd > diff y.  
pd 'show'  
)
```

3 自己相関

時系列とは、一定の等間隔にとらえられた時間間隔毎に並んだ観測値の列であり、経済や工業、医学を初めとし、様々な分野で観測されている。気温、GDP、貨幣供給量、株価、計器の圧力、脳波、etc.

3.1 自己相関 ACF(k) autocorreration function

ACF(K)		
--------	--	--

ある確率過程が定常であるならば、時間の経過での変化の関係をとらえることによって、時間領域での性質を要約することができる。

時系列 y_n と時刻を k だけシフトした y_{n-k} を時系列 y_n の自己共分散関数 (autocovariance function) といい、時間が y_{n-k} だけ経過したとき、確率変数がどのように経過するかを示している。

自己相関係数 (autocorrelation) は、標準偏差で規格化された自己共分散であり、自己共分散、自己相関関数は時系列の過去の変動との関連の強さを表現したものである。

ここでは自己共分散関数と自己相関関数を求める。

自己共分散関数は `acf2 y.` を求めたあと `C4` とタイプすれば各階差毎に表示される。

3.1.1 解説

$$ACF(k) = \frac{\sum_{t=k+1}^n (Y_t - \bar{Y})(Y_{t-k} - \bar{Y})}{\sum_{t=1}^n (Y_t - \bar{Y})^2}$$

ACF では \bar{Y} に原系列の平均を用いる。Paerson ではずらした後の各個数の平均を用いる。

定常過程の場合、 $X(t)$ の期待値は、時刻 t に依存しない。自己共分散 $Cov[X(t_1), X(t_2)]$ も時刻 t_1, t_2 には依存しないで、その時間差 $\tau = t_1 - t_2$ のみに依存する。

次のスクリプトは、自己相関関数を一度に求めてしまうものである。

3.1.2 経過と説明

a

3 6 8 4 4 8

C1=: (<\. y.) ,. |.@(<\)y.

右の欄は上から $Y_t, Y_{t-1}, Y_{t-2}, Y_{t-3}...$ となっている。

左の欄は頭を落とし、右の欄にサイズを合わせたもの。

C1

```
+-----+-----+
|3 6 8 4 4 8|3 6 8 4 4 8|
+-----+-----+
|6 8 4 4 8 |3 6 8 4 4 |
+-----+-----+
|8 4 4 8   |3 6 8 4   |
+-----+-----+
|4 4 8     |3 6 8     |
+-----+-----+
|4 8       |3 6       |
+-----+-----+
|8         |3         |
+-----+-----+
```

mean y.

5.5

各要素から平均を引く

C3 NB. C1 - (L:0) mean y.

```
+-----+-----+
```

```

|_2.5 0.5 2.5 _1.5 _1.5 2.5|_2.5 0.5 2.5 _1.5 _1.5 2.5|
+-----+-----+
|0.5 2.5 _1.5 _1.5 2.5    |_2.5 0.5 2.5 _1.5 _1.5    |
+-----+-----+
|2.5 _1.5 _1.5 2.5        |_2.5 0.5 2.5 _1.5        |
+-----+-----+
|_1.5 _1.5 2.5            |_2.5 0.5 2.5            |
+-----+-----+
|_1.5 2.5                  |_2.5 0.5                  |
+-----+-----+
|2.5                        |_2.5                        |
+-----+-----+

```

左右の各列ごとに左右の要素を掛け合わせ列（次数）ごとに合計する。

```
C4 =: +/ "1 */ "2 >C3 NB. Cov
```

```
23.5 _5.25 _14.5 9.25 5 _6.25
```

```
+/L:0 ^&2>{.{.C3 NB. Var
```

```
acf2 a NB. Cov % Var
```

```
1 _0.223404 _0.617021 0.393617 0.212766 _0.265957
```

3.1.3 SCRIPT

```
acf2=: 3 : 0
```

NB. ACF autocorellation coefficients (many times at once)

```
C1=: ( <\. y.) ,. |.@(<\)y.
```

```
C2=: (mean=. +/ % #) y.
```

```
C3=: C1 - L:0 C2
```

```
C4=: +/ "1 */ "2 >C3
```

```
NB. C5=:> +/ L:0 ^&2 L:0 {."1 C3
```





図 5: ACF

C4 % (+/ ^&2 >{. { . C3)
)

3.1.4 出典

Stephen A. DeLurgio [Forecasting Principles and Applications] McGRAW-HILL 1998

3.2 C.A. Reiter による自己相関のスク립ト

	<pre> acov=: ([: +/ [((-@[}.]) * }.) (- (+/ % #))@[]) % #@[NB. r(t) acor=: (acov % 0: acov]) " 0 1 NB. p(t) </pre>
--	--

自己相関係数の推定値 $\hat{\rho}(\tau) = \frac{\hat{\gamma}(\tau)}{\hat{\gamma}(0)}$ により求められる。

C.Reiter による ACF のスク립ト (結果は同一)

Script is written by Criford A. Reiter [acor] is same as ACF

3.2.1 解説

$$\hat{\gamma}(t) = \frac{1}{T} \sum_{i=1}^{T-t} (X_{t+i} - \bar{X})(X_i - \bar{X})$$
$$\rho_{ss} = \frac{R_{ss}(s)}{R_{ss}(0)}$$

3.2.2 経過と説明

```
2 acor a
_0.617021
```

```
3 acor a
0.393617
```

```
acf2 a
1 _0.223404 _0.617021 0.393617 0.212766 _0.265957
```

3.3 偏自己相関係数 Pearsons autocorrelation coefficient PACF

PACF partial autocovariance function

PARCOR partial autocorrelation coefficient

3.3.1 解説

$$r_{Y_t Y_{t-k}} = \frac{COV(Y_t, Y_{t-k})}{S_{Y_t} S_{Y_{t-k}}}$$

自己相関係数が、時系列における時点間の関係の強さを示すものであるのに対し、偏自己相関係数は、与えられたラグの期間内の影響を取り除いた時の関係の強さを示す。モデルの次数を示すときに用いられる。

3.3.2 経過と説明

```
a=:3 6 8 4 4 8
```

共分散 $Cov(Y_t Y_{t-k})$

```
C3=: (+/ "1 */ "2 > dev L:0 C1) % n=:> {. "1 <:@# L:0 C1
23.5 _4 _14 9.33333 6 0
```

$S_{Y_t} S_{Y_{t-k}}$ は Y_t , Y_{t-k} の標準偏差

```
pacf a
1 _0.25 _0.911322 0.802955 1 0
```




図 6: PACF

3.3.3 Script

```
pacf=: 3 : 0
NB. Pearsons autocorrelations (many times at once)
C1=: ( <\. y.) ,. |.@(<\)y.
dev=. - (+/ % #)
NB. Cov(Yt Yt-k)
C3=: (C2=:+/ "1 */ "2 > dev L:0 C1) % n=:> {. "1 <:@# L:0 C1
NB. Syt Syt-k
C4=: +/ L:0 *: L:0 dev L:0 C1
C5=:*/ "1 > %: L:0 C4 % L:0 n2=:<:@# L:0 C1
```

C3 % C5

)

3.3.4 出典

Stephen A. DeLurgio [Forecasting Principles and Applications] McGRAW-HILL 1998

4 Autoregression 自己回帰

時系列モデルを用いると時系列の特徴を簡潔に表現できる。ここでは自己回帰モデル (autoregressive model) を当てはめる問題を取り扱う。

AR (Autoregression) の解法には Yule-Walker 法, バーク法, ハウスホルダー法がある。

4.1 Yule-Walker 法

$$C_M a_M = c_M$$

$$c(k) = \frac{1}{N} \sum_{s=1}^{N-k} x(s)x(s+h) \quad (k = 0, 1, 2, \dots, M)$$

$$C_M = \begin{bmatrix} c(0) & c(1) & \dots & c(M-2) & c(M-1) \\ c(1) & c(0) & c(1) & \dots & c(M-2) \\ \vdots & \vdots & \ddots & \vdots & \vdots \\ c(M-2) & \dots & c(1) & c(0) & c(1) \\ c(M-1) & c(M-2) & \dots & c(1) & c(0) \end{bmatrix} \begin{bmatrix} a_1 \\ a_2 \\ \vdots \\ \vdots \\ a_p \end{bmatrix} = \begin{bmatrix} C_1 \\ C_2 \\ \vdots \\ \vdots \\ C_p \end{bmatrix}$$

$$c_M = (c(1), c(2), \dots, c(M))^T$$

自己相関 ACF(k) を用いて C_M を作れば, この方程式はクラメル法やガウス法で解くことができ Yull-Walker 法はこの yull-walker 方程式を直接解くことにより AR 係数の推定値を求められる。(レビンソンのアルゴリズムは用いない。)

4.1.1 経過と結果

```
acf2 a
1 _0.223404 _0.617021 0.393617 0.212766 _0.265957
```

Yull-walker 方程式 acf2 の結果を 3 次の C_M に、右に c_M を連結したもの

1	_0.223404	_0.617021	_0.223404
_0.223404	1	_0.223404	_0.617021
_0.617021	_0.223404	1	0.393617

上の拡大行列をクラメール法で解いたもの。

```
cr=: %.):"1          NB. Cramer Method
```

```
3 yw a
      1 _8.88178e_16 6.66134e_16  _0.37704
_2.22045e_16          1 4.44089e_16  _0.70024
_6.66134e_16 _8.32667e_16          1 0.00453847
```

$$X(t) = -0.37704x(t-1) - 0.70024x(t-2) + 0.00453847x(t-3)$$

4.1.2 Script

```
cmtest=:4 : 0
NB. Yule- Walker sub
NB. make C(m) Matrix
NB. for univariate AR
T1=: (|.}. M) , M=: y.
T2=: x. # ,: T1
T3=: ({. TMP), <: 1{ TMP=: $ T2
T4=: (-x.) {. ("1) TMP2=: T3 $ , T2
)

yw=:4 : 0
NB. Yule Walker(Univariate)
```

```

NB. Data type is Yoko list
NB. X(t)=X1(t-1)+ X2(t-2) + X3(t-3).....
NB. Usage: x. jisuu // y. Data
Y1=: acf2 y.
Y2=:(>: x.) {. Y1
Y3=:}. (>: x.) {. Y1
Y4=: (x. cmtest Y2),. Y3
Y5=: (cr=: %.):"1) Y4
)

```

4.2 AIC

AR の赤池情報量基準は次の式による。最適次数は AIC の最小値を選択すればよい。

$$\begin{aligned}
 AIC_m &= -2l(\hat{\theta}) + 2(\text{パラメータ数}) \\
 &= N(\log 2\pi\hat{\sigma}_m^2 + 1) + 2(m + 1)
 \end{aligned}$$

```

2 aicctest d1
3.76929

NB. AR 1JI
NB. *****
aicctest=: 4 : 0
NB. find AIC value
NB. Usage: x. aicctest y. // x. is jisuu // y. is data
y=: dev y.
A0=:}: A00=:x. <\ y NB.
A1=: {"1 x. yw y NB. tmp is reg
A2=: (tmp0=:|"1 >A0) +/ . * A1 NB. (|"1 ) is rotate A0
A3=: (tmp1=: (x. }. y)) - A2 NB.
A4=: (tmp2=: +/ *: A3) % n=: (# A3) - x. NB. tmp2 is Q
A5=: (n * ^ . A4)++:x.
)

```

4.3 回帰と予測

回帰と予測は, 次のような組み合わせを作ると $x(t-4)x(t-3)x(t-2)x(t-1)$ になっており, 回帰は, これに, $A(m)$ を掛ける。最下行は, 1 期先の予測に用いる。

```
|"1 > 4 <\ a=. >: i. 12
 4 3 2 1 | 5
 5 4 3 2 | 6
 6 5 4 3 | 7
 7 6 5 4 | 8
 8 7 6 5 | 9
 9 8 7 6 | 10
10 9 8 7 | 11
11 10 9 8 | 12
12 11 10 9 | x
```

```
29 pred corn
20.0508
```

```
plot_ar =: 4 : 0
NB. plotAR
x. aicctest y.
pd 'reset'
pd 'color green'
pd dev y.
pd 'color red'
pd (x. # 0), A2
pd 'show'
)
```

```
pred=: 4 : 0
NB. prediction AR
NB. Usage: x. pred y.
```



```
x. aicctest y.  
A09=: (+/ % #) y.  
A10=: |. {: > A00  
A11=: A10 +/ . * A1  
A12=: A09 + A11  
)
```

4.3.1 参考文献

石黒真木夫「予測と AR モデル」(尾崎・北川編 時系列解析の方法 朝倉書店) 19
98

5 レビンソンのアルゴリズム

レビンソンのアルゴリズムは最初に偏相関係数を求め、これをもとに回帰係数を計算する方法である。一変数の場合には、 J は直接ユールウオーカー方程式を解くことができるが、ここでは、多変量自己回帰係数を求めるアルゴリズムの理解のために、作成した。

5.1 一変数

[0 次]

$$K_0 = C_0$$

[1 次]

$$K_1 = \frac{C_1}{V_0}$$

$$\text{reg}_1 = K_1$$

[2 次]

$$K_{2(2)} = \frac{1}{V_1}(C_2 - K_1 C_1)$$

$$K_{2(1)} = K_{1(1)} - K_{2(2)} K_{1(1)}$$

$$\text{reg}_2 = K_{2(1)} + K_{2(2)}$$

$$V_2 = V_1(1 - (K_{2(2)})^2)$$

[3 次]

$$K_{3(3)} = \frac{1}{V_2} \left(C_3 - \sum \begin{pmatrix} K_{2(1)} C_2 \\ K_{2(2)} C_1 \end{pmatrix} \right)$$

$$K_{3(1)} = K_{2(1)} - K_{3(3)} K_{2(2)}$$

$$K_{3(2)} = K_{2(2)} - K_{3(3)} K_{2(1)}$$

$$\text{reg}_3 = K_{3(1)} + K_{3(2)} + K_{3(3)}$$

$$V_3 = V_2(1 - (K_{3(3)})^2)$$

[4 次]

$$K_{4(4)} = \frac{1}{V_4} \left(C_4 - \sum \begin{pmatrix} K_{3(1)} C_3 \\ K_{3(2)} C_2 \\ K_{3(3)} C_1 \end{pmatrix} \right)$$

$$K_{4(1)} = K_{3(1)} - K_{4(4)} K_{3(3)}$$

$$\begin{aligned}
K_{4(2)} &= K_{3(2)} - K_{4(4)}K_{3(2)} \\
K_{4(3)} &= K_{3(3)} - K_{4(4)}K_{3(1)} \\
\text{reg}_4 &= K_{4(1)} + K_{4(2)} + K_{4(3)} + K_{4(4)} \\
V_4 &= V_3(1 - (K_{4(4)})^2)
\end{aligned}$$

右の欄は分散 σ^2 。回帰係数は、上から 0 次, 1 次.. 5 次 (=x.) まで同時に求めている。
AR の係数は、次数に応じ、各行に表示される。右端の列は、分散 σ^2 である。

C
9.33354 7.87931 4.50071 0.512286 _2.90688 _4.87683 _5.10031 _3.82648

5 levin C

0	0	0	0	0	9.33354
0.844193	0	0	0	0	2.68189
1.52126	_0.802026	0	0	0	0.956775
1.50865	_0.778118	_0.0157164	0	0	0.956538
1.50777	_0.821906	0.0691834	_0.0562752	0	0.953509
1.51126	_0.826198	0.120175	_0.149817	0.0620402	0.949839

yull-walker 法で求めた結果と合致している。

3 levin acf2 d1

0	0	0	1
_0.223404	0	0	0.950091
_0.380226	_0.701965	0	0.481928
_0.37704	_0.70024	0.00453847	0.481918

NB. this line is AR

$$X(t) = -0.37704x_{t-1} - 0.70024x_{t-2} + 0.00453847x_{t-3}$$

3 yw d1

1	_8.88178e_16	6.66134e_16	_0.37704
_2.22045e_16		1 4.44089e_16	_0.70024

_6.66134e_16 _8.32667e_16

1 0.00453847

5.1.1 参考文献

廣田薫・生駒哲一「確率過程の数理 (数理工学基礎シリーズ)」朝倉書店 2001

5.2 Householder 法

自己回帰 Autoregression	aregb [回帰係数を求める] areg [平均値・決定係数・AIC 値・回 帰係数]	m aregb n m areg n m 次数 n Data
------------------------	--	---

Householder による AR ARMA は鈴木教授がスクリプトを公開しておられるので, 簡単に再掲した。

```
3 aregc tmp3
750.25 0.946196 341.705 1.37654 _0.236394 _0.221991
```

平均値 決定係数 AIC 値 回帰係数

Yule-Walker 法との比較

```
3 yw tmp3
          1 _1.01918e_13 _1.01252e_13 1.09991
1.84741e_13          1 2.22045e_13 0.0260682
_1.05693e_13 _1.24345e_13          1 _0.238416
```

Householder $f(x) = 1.37654X_{t-1} - 0.236394X_{t-2} - 0.221991X_{t-3}$

Yule-Walker $f(x) = 1.09991X_{t-1} + 0.0260682X_{t-2} - 0.238416X_{t-3}$

NB. n=48 monthly values of a Common Stock Price

NB. data from DeLurgio SERIESB.DAT

5.2.1 Script

NB. =====

NB. Auto Regression by G. suzuki

```

NB. =====
adep=: [}.]-(+/%#)@]
aindep=: 4 : 0
|:(0,<:x.)}.(x.),<:#y.)$y.-(+/%#)y.
)

aregb=:([adep])%.[aindep]
aregc=: 4 : 0
y=: (k=:x.)}.t=:y. - m=:(+/%#)y.
x=: |:(0,<:k)}.(k,<:n=:#t)$t
q=:+/*:y-x+/ . * b=:y%.x
c=:1-q%+/*:y-(+/%#)y
m,c,((n^q%n=:n-k)++:k),b
NB. ave r2 AIC ar
)

```

6 自己回帰移動平均 : ARMA

鈴木教授の ARMA のスクリプトの再掲

```
NB. =====
NB. ARMA
NB. script written by Suzuki , 30 Sept. 2000
NB. =====

mav=:(([:+/[+/\"_1{:,:}.y.)@])%2:*[]'(+/\%[]@.(2:|[]
sac=: 3 : 0
NB. Seasonal Adjustment Coefficient
y=. (+/12+/\"_1{:,:}.y.)%24 NB. moving average
(+/%#) ((<:(#y.)%12),12)$({6}. _6 }.y.) %y
)

aregm=: 4 : 0
NB. autoregressive model
b=. (p=. x.}.t)%q=.}: x. ]\t=.y.-m=(+/%#)y.
c=.1-(r=.+/*:p-q+/ . * b)%+/*:p
m,c,(((#t)*^.r%#p)++:x.),b
)

trend=: 4 : 0
b=. x. (}.%.[ : } : ]\t=. y-m=(+/%#)y=.}.y.
'r t'=. '' ;(-x.){.t
while. (#r)<{.y.
  do. r=. r, m+q [t=. (}. t),q=.+/b*t
end.
)
```

```

predict=: 4 : 0
NB. prediction by ARMA MODEL
z=.(+/12+/\ "_1({: ,: }.y=.}. y.) % 24
s=.(+/%#)((<:(#y)%12),12)$(6}. _6}. y)%z
(k$s)*x. trend (k=. {. y.),z
)

```